
JAGS script:
###Model correcting for Differential Verification Bias###
model
{
	###Prior distributions###
	phi ~ dbeta(1,1)
	ST ~ dbeta(1,1)
	CT ~ dbeta(1,1)
	SR1 ~ dbeta(151.125,3.875)	# Sensitivity range R1: 95-100%
	CR1 ~ dbeta(151.125,3.875)	# Specificity range R1: 95-100%
	SR2 ~ dbeta(31.5,3.5)		# Sensitivity range R2: 80-100%
	CR2 ~ dbeta(58.1,24.9)		# Specificity range R2: 40-60%

	VT1R1 ~ dunif(0,1)
	VT0R1 ~ dunif(0,1)
	VT1R2 ~ dunif(0,1)
	VT0R2 ~ dunif(0,1)

	###Probability data###
	p1 <- phi*ST+(1-phi)*(1-CT)

	p2 <- (SR1*phi*ST/(phi*ST+(1-phi)*(1-CT))+(1-CR1)*(1-phi)*(1-CT)/(phi*ST+(1-phi)*(1-CT)))
	p3 <- (SR1*phi*(1-ST)/(phi*(1-ST)+(1-phi)*CT)+(1-CR1)*(1-phi)*CT/(phi*(1-ST)+(1-phi)*CT))

	p4 <- (SR2*phi*ST/(phi*ST+(1-phi)*(1-CT))+(1-CR2)*(1-phi)*(1-CT)/(phi*ST+(1-phi)*(1-CT)))
	p5 <- (SR2*phi*(1-ST)/(phi*(1-ST)+(1-phi)*CT)+(1-CR2)*(1-phi)*CT/(phi*(1-ST)+(1-phi)*CT))

	###Likelihood observed data###
	#Stage 1
	T1 ~ dbin(p1,n)

	#Verification Stage
	nT1R1 ~ dbin(VT1R1,T1)
	n0 <- n-T1
	nT0R1 ~ dbin(VT0R1,n0)
	nT1R2 ~ dbin(VT1R2,T1)
	nT0R2 ~ dbin(VT0R2,n0)

	#Stage 2
	T1R11 ~ dbin(p2,nT1R1)
	T0R11 ~ dbin(p3,nT0R1)
	T1R21 ~ dbin(p4,nT1R2)
	T0R21 ~ dbin(p5,nT0R2)

	#Predictive values
	PPVT <- ST*phi/(ST*phi+(1-CT)*(1-phi))
	NPVT <- CT*(1-phi)/(CT*(1-phi)+(1-ST)*phi)

	#Accuracy measures T with respect to Reference test 1
	STR1 <- p2*p1/(p2*p1+p3*(1-p1))
	CTR1 <- (1-p3)*(1-p1)/((1-p2)*p1+(1-p3)*(1-p1))
	PPVTR1 <- p2
	NPVTR1 <- 1-p3

	#Accuracy measures T with respect to Reference test 2
	STR2 <- p4*p1/(p4*p1+p5*(1-p1))
	CTR2 <- (1-p5)*(1-p1)/((1-p4)*p1+(1-p5)*(1-p1))
	PPVTR2 <- p4
	NPVTR2 <- 1-p5
}





